

Statistics 147 In Class Exercise #3

Summer 2020; 10 pts

NAME: _____ ID: (last 4 #s only) _____

GOAL: (In class practice) Getting more acquainted with **do loops** and **subsetting if** statements in **SAS**.

NOTE: You will need to download the following data file from iLearn (under DataFiles): **dograces_su19.dat**

Invoke **SAS**.

1. Luke, Rachel and Bentley have taken up dog-sled racing in hopes that someday they can enter The Iditarod Dog Sled Race in Alaska with *Trusty Dusty*, *White Shadow* and *Lakota Dakota* as their respective lead dogs. Since there is a lack of snow in Southern California, each have obtained a sled in which the runners have been replaced by wheels. After a significant number of practice runs, Luke, Rachel and Bentley race every day (not necessarily together) for 3 months and record their time to finish the course (in minutes). Ruihan does not believe there is a significant difference in mean finishing times between the three teams. To test this claim, Ruihan takes three independent random samples of 10 days times for each of the three teams, yielding the following data:

```
Filename: dograces.dat
Dusty Shadow Lakota
45.5  43.6  64.9
59.2  59.9  66.2
38.4  39.8  56.5
68.8  70.4  75.7
51.9  50.2  55.2
47.4  48.9  61.5
41.6  40.3  49.9
58.9  58.0  63.5
60.7  60.9  65.0
47.0  46.5  62.6
```

NOTE:

- ♠ The data is located in a datafile named **dograces.dat**.
- ♠ The headings are included in the data file. The actual data begins on line 3.
- ♠ **Assume** the *Trusty Dusty* is dog team leader 1, *White Shadow* is dog team leader 2, *Lakota Dakota* is dog team leader 3.
- ♠ μ_i = true average finishing time for dog team i and σ_i^2 = variance of the finishing time for dog team i , $i = 1, 2, 3$

(i) Read in and print out the data using nested Do loops. (Be sure to give your columns the appropriate dog team leader names!) (4 pts)

We write the following code:

```
/* Set up format of the output */
options nocenter ps = 55 nocenter ls = 78 nodate nonumber formdlim='*';
/* ls = linesize, ps = pagesize
    nocenter      justifies the output so it is not centered on the page
    nodate        suppresses printing of today's date on each page of output
    nonumber      suppresses printing of page number on each page of output
```

```

formdlim          overrides the internal page breaks and replaces them
                   with the designated symbol*/

/* Use DM to clear all windows except the editor window */

DM log "odsresults; clear; out; clear; log; clear;";

ods graphics off;

/* Create a temporary SAS data set */
data alldogs;

    /* Use an infile statement to open the data file.
       Use firstobs command to indicate the actual data starts on Line 3 of the data file.*/
    infile 'c:\Luke\summer2020\sul9147\datafiles\dograces_sul9.dat' firstobs = 3;
    /* BE SURE TO CHANGE THE PATH TO THE DATAFILE */

    /* Use nested Do loops to read in and output the data */
    /* Do loops for the 10 rows of data */
    do rows = 1 to 10;
        /* Do Loop for the three teams of dogs */
        do team = 1 to 3;
            /* Use if-then-else structure to name the team leaders */
            if      team = 1 then leader = 'Trusty Dusty ';
            else if team = 2 then leader = 'White Shadow ';
            else      leader = 'Lakota Dakota';
            /* Use input statement to read the data obs from the file*/
            input time @@;
            /* then use the output statement to write the values to
               the temp SAS dataset 'alldogs' */
            output;
        /* Close the team loop */
        end;
    /* Close the rows loop */
    end;
run;
/* Print the data */
proc print data = alldogs;
    /* Set up titles */
    title1 'Statistics 147 In Class Exercise #3';
    title2 'Summer 2020';
    title3 'Your name goes here';
    title4 'SAS Question 1';
run;
quit;

```

Save your program file as **inclass3_147_su19**. Execute the file. When your output appears on the screen, have Luke, Ruihan or your neighbor initial here. _____

(ii) Sort the teams by the dog leader names. (2 pts)

This can be accomplished by adding the following lines of code right **before** the **quit** statement.

```

/* Sort by leader */
proc sort;
    by leader;
run;
proc print;
/* Revise title5 */
    title5 'Part (ii)';
run;

```

Save and execute your program. When your output appears on the screen, have Luke, Ruihan or your neighbor initial here. _____

(iii) Generate the mean, standard deviation and variance for each of the three dog teams. (3 pts)

This can be accomplished by adding the following lines of code right **before** the **quit** statement.

```

/* Use proc means to generate required descriptive statistics */
proc means mean stddev var;
    /* Revise title5 */
    title5 'Part (iii)';
    by leader;
    var time;
run;

/* Or can use proc univariate */
proc univariate;
    /* Revise title5 */
    title5 'Part (iii)';
    by team;
    var time;
run;

```

Save and execute your program. Complete the following.

leader	Mean	Standard Deviation	Variance
<i>Lakota Dakota</i>			
<i>Trusty Dusty</i>			
<i>White Shadow</i>			

(iv) Create a new SAS dataset, called **onlyDusty**, and bring in the data using the SET command. Use the appropriate IF structure to restrict the data to the **Trusty Dusty** dog team. Be sure to print the data! (2 pts)

This can be accomplished by adding the following lines of code right **before** the **run** statement.

```

/* Create new temporary SAS dataset to restrict attention to Trusty Dusty */
data onlyDusty;
    /* Use set command to bring in all the data from the SAS dataset alldogs */
    set alldogs;

```

```

        /* Use if statement to restrict the data to team 1 (Trusty Dusty) */
        if leader = 'Trusty Dusty ';
        /* Or can use
           if team = 1; */
run;

/* Print the results */
proc print data = onlyDusty;
    /* Revise title5 */
    title5 'Part (iv)';
run;

```

Save and execute your program. When your output appears on the screen, have Luke, Ruihan or your neighbor initial here. _____

(v) Create a new SAS dataset, called **bothShadCody** and bring in the data using the SET command. Use the appropriate IF structure to restrict the data to the **White Shadow** and **Lakota Dakota** dog teams. Be sure to print the data! (2 pts)

This can be accomplished by adding the following lines of code right **before** the **run** statement.

```

/* Create new temporary SAS dataset to restrict attention to White Shadow and Lakota Dakota */
data bothShadCody;
    /* Use set command to bring in all the data from the SAS dataset alldogs */
    set alldogs;

    /* Use if statement to restrict the data to team 2 (White Shadow) and team 3 (Lakota Dakota) */
    if leader = 'White Shadow ' or leader = 'Lakota Dakota';
    /* Or can use
       if team = 2 or team = 3; */
run;

/* Print the results */
proc print data = bothShadCody;
    /* Revise title5 */
    title5 'Part (v)';
run;

```

Save and execute your program. When your output appears on the screen, have Luke, Ruihan or your neighbor initial here. _____

You have now successfully completed **In Class SAS Practice #3**. Please turn in this worksheet. Be sure to log off your account, take your flash drive (if you used one) and make sure that your work area is neat and clean. Have a nice day!

Luke & Ruihan